

```

UM Kurve2 :n
STARTEZUFALL -280 280
BILD LÖSCHTEXT VERSTECKTIGEL
TEIL [ "x "y ]
SETZE "x [] SETZE "y []
Poly :n
SETZE "h 0,05
WIEDERHOLE 30 [ // 30 Wiederholungen. //
FARBEIGEL ELEMENT :Farben (REST WHZAHL 15 ) + 1
SETZE "xxm [] SETZE "yym []
// Man berechnet die neuen Koordinaten der extremen Punkte der Listen :x und :y um Berechnungen zu vereinfachen. Man
kann jetzt diese zwei Listen je nach der Wiederholungen abstumpfen, von wo Beschleunigung der Vorzeichnung. //
  SETZE "xd ( LETZTES :x ) + ( :h * ( ( ERSTES :x ) - ( LETZTES :x ) ) )
  SETZE "yd ( LETZTES :y ) + ( :h * ( ( ERSTES :y ) - ( LETZTES :y ) ) )
  SETZE "xxm SATZ :xxm ( ERSTES :x ) + ( :h * ( ( ELEMENT :x 2 ) - ( ERSTES :x ) ) )
  SETZE "yym SATZ :yym ( ERSTES :y ) + ( :h * ( ( ELEMENT :y 2 ) - ( ERSTES :y ) ) )
  SEGMENT [ ELEMENT :x 2 ELEMENT :y 2 ] [ LETZTES :xxm LETZTES :yym ]
  SEGMENT [ :xd :yd ] [ LETZTES :xxm LETZTES :yym ]
  SETZE "x OHNEERSTES :x SETZE "y OHNEERSTES :y
WIEDERHOLE ( :n - 2 ) [ // die vorhergehende Kurve verfolgt das folgende.//
  SETZE "xxm SATZ :xxm ( ERSTES :x ) + ( :h * ( ( ELEMENT :x 2 ) - ( ERSTES :x ) ) )
  SETZE "yym SATZ :yym ( ERSTES :y ) + ( :h * ( ( ELEMENT :y 2 ) - ( ERSTES :y ) ) )
  SEGMENT [ ELEMENT :x 2 ELEMENT :y 2 ] [ LETZTES :xxm LETZTES :yym ]
  // Man zieht hier. //
  SETZE "x OHNEERSTES :x SETZE "y OHNEERSTES :y
]
  SETZE "xxm SATZ :xxm :xd SETZE "yym SATZ :yym :yd
  SETZE "x [] SETZE "y []
WIEDERHOLE :n [ // Man aktualisiert die Liste der neuen Koordinaten.//
  SETZE "x SATZ :x ERSTES :xxm
  SETZE "y SATZ :y ERSTES :yym
  SETZE "xxm OHNEERSTES :xxm
  SETZE "yym OHNEERSTES :yym
]
]
ENDE

UM Poly :n
SCHREIBLINIE [ Berechnung der Anfangspunkte. ]
WIEDERHOLE :n [
  STARTEZUFALL -260 260
  SETZE "x SATZ :x ZUFALLZAHL
  SETZE "y SATZ :y ZUFALLZAHL
  STARTEZUFALL 0 0,005 WART ZUFALLZAHL
]
ENDE

setze "Farben [ weiss grün gelb rot orange linde pink sarcelle braun rosa violett gelb grau schwarz blau olivefarbe ]
Kurve2 6

```